

电子计算机常用算法选编

电子计算机常用算法选编

杨　洪　编

山东科学技术出版社
一九八五年·济南

内 容 简 介

本书是一部计算机常用算法的工具书。它是用我国计算机使用较普遍的ALGOL60语言书写的，文中汇编了部分科技计算和工程设计中常用的算法，并包括部分近年来出现的应用数学新方法。

全书选入了四十七个算法标准过程。内容分为五个部分：一般数学方法，回归分析，数据处理与插值，聚类分析，线性规划与图论。

本书可供使用计算机的科技人员、工程设计人员查阅使用，也可供理工科高等院校师生作教学参考书。

电子计算机常用算法选编

杨 洪 编

*

山东科学技术出版社出版
山东省新华书店发行
山东新华印刷厂印刷

*

737×1092毫米32开本 9.75印张 208千字
1985年3月第1版 1985年3月第1次印刷
印数，1—20,000

书号 13195·128 定价 2.00元

前　言

自从算法语言于五十年代末问世以来，不仅为专业程序计算人员提高工作效率提供了有利的条件，同时也为通用数字电子计算机计算程序的相互交流带来了很大的方便，算法语言的广泛应用推动了通用数字电子计算机在科学研究、生产技术、国防建设、文化教育等方面的广泛普及应用。

ALGOL语言是目前我国电子计算机当中广泛被使用的程序自动化语言之一，将一些典型计算方法按照ALGOL语言的规定编制成具有通用性、又有灵活功能的单用过程(procedure)或函数过程，并对它的功能、计算原理、使用方法作适当地介绍，对于计算机的使用者缩短编写程序的时间，避免重复性劳动，节省使用计算机的机时等方面都是十分有利的。本书试图为读者提供部分专用过程，使用者只需根据本人的计算任务去选择有关的过程，嵌入自己的程序中去使用。

本书各段对相应的数学方法作了简要说明，但是我们着重于使读者了解每一个专用过程的不同功能，并准确地掌握其使用方法，以便用它得到正确的计算结果。为了帮助读者掌握使用方法，除对每一算法的所有参数作了详尽说明外，在后面都举了一至二个实例以供参考。

本书是按照DJS—6机的各种规定编写的。DJS—6机的ALGOL语言基本符合ALGOL60语言文本的规定，因而它不仅可以直接用于DJS—6机，只要略加改动就可以在配置有ALGOL语言软件的其它电子计算机上使用。

ALGOL 语言与PASCAL语言的规定比较接近。熟悉PASCAL语言的读者，将本书提供的程序改写为PASCAL过程（procedure）将不会感到困难。

本书介绍的全部程序均在DJS—6电子计算机或DJS—21电子计算机上调试通过，其例题都是在计算机上经过检验并取得了数值结果的。读者可以将自己准备选用的专用过程按各节后面例题程序的格式送入电子计算机去检验其正确性，然后再将其嵌入自己的主程序中去使用。

本书的线性规划与图论方面的程序设计过程曾得到山东大学郑汉鼎老师的指导，并得到刘家壮、张天赐同志的热情帮助，在此表示衷心的感谢。

编　者

1984.7.27

目　录

第四部分 聚类分析方法

第五部分 线性规划与图论

附　录

第一部分　一般数学方法

1.1　复数的基本运算

一、功能

本过程可进行复数的加法、减法、乘法、除法、开平方运算．调用过程一次，执行上述五种运算当中的一种．

二、计算步骤

在处理复数域的计算问题时，会遇到大量的复数加、减、乘、除、开平方等的混合运算，使用本过程为这一类程序的编写带来许多方便．在过程中，复数的实部与虚部分别存放在一个数组$Z(1:2)$的两个单元中．随着控制参数N的不同取值，每次调用过程时，只执行上述五种基本运算之一．

当$N=1$　　$Z_1+Z_2\Rightarrow Z_3$；

$N=2$　　$Z_1-Z_2\Rightarrow Z_3$；

$N=3$　　$Z_1*Z_2\Rightarrow Z_3$；

$N=4$　　$Z_1/Z_2\Rightarrow Z_3$；

$N=5$　　$\sqrt{Z_1}\Rightarrow Z_2$，　$-\sqrt{Z_1}\Rightarrow Z_3$．

三、过程参数说明

过程名称　$MM(N, Z_1, Z_2, Z_3)$；

N：控制参数．$N=1$，加法；$N=2$，减法；$N=3$，乘法；$N=4$，除法；$N=5$，开平方，整型值参．

Z_1，Z_2，Z_3〔1：2〕：参加运算的复数数组，计算结果复数数组．Z_1〔1〕、Z_2〔1〕、Z_3〔1〕分别存放复数实部，Z_1〔2〕、Z_2〔2〕、Z_3〔2〕分别存放复数虚部，实型数组。

四、过程

```
'PROCEDURE'MM(N,Z1,Z2,Z3);
'VALUE'N; 'INTEGER'N; 'ARRAY'Z1,Z2,Z3;
'BEGIN' 'REAL'X,Y; 'SWITCH'SW:=V0;
  'SWITCH'T:=V1,V2,V3,V4,V5;
  'GOTO'T[N];
  V1: Z3[1]:=Z1[1]+Z2[1]; Z3[2]:=Z1[2]+Z2[2];
      'GOTO'V0;
  V2: Z3[1]:=Z1[1]-Z2[1]; Z3[2]:=Z1[2]-Z2[2];
      'GOTO'V0;
  V3: Z3[1]:=Z1[1]*Z2[1]-Z1[2]*Z2[2];
      Z3[2]:=Z1[1]*Z2[2]+Z1[2]*Z2[1]; 'GOTO'V0;
  V4: 'IF'ABS(Z2[1])'GQ'ABS(Z2[2])'THEN'
      'BEGIN'X:=Z2[2]/Z2[1];Y:=Z2[1]+X*Z2[2];
        Z3[1]:=(Z1[1]+Z1[2]*X)/Y;
        Z3[2]:=(Z1[2]-Z1[1]*X)/Y; 'END''ELSE'
      'BEGIN'X:=Z2[1]/Z2[2];Y:=Z2[2]+X*Z2[1];
        Z3[1]:=(Z1[1]*X+Z1[2])/Y;
        Z3[2]:=(Z1[2]*X-Z1[1])/Y; 'END';'GOTO'V0;
  V5: X:=SQRT(Z1[1]↑2+Z1[2]↑2); X:=SQRT(X);
      Y:='IF'Z1[1]=0.0'THEN'1.57079633*SIGN(Z1[2])
        'ELSE'ARCTAN(Z1[2]/Z1[1])+1.57079633
        *(1-SIGN(Z1[1]));
      Z2[1]:=X*COS(Y/2); Z2[2]:=X*SIN(Y/2);
      Z3[1]:=-Z2[1]; Z3[2]:=-Z2[2];
  V0: 'END';
```

五、例题

计算　(2+2*i*)+(2−2*i*)　结果 4+0*i*

(2+2*i*)−(2−2*i*)　0+4*i*

(2+2*i*)*(2−2*i*)　8+0*i*

$$(1.5_{10}10+1_{10}20i)/(2_{10}18+1_{10}10i)$$

$$=0.2575_{10}-6+0.5_{10}2i$$

$$\sqrt{1.264_{10}18+1.548_{10}18i}$$

$$=\pm(1.27720387_{10}9+6.06011319_{10}8i)$$

程序

```
'BEGIN'
   'ARRAY'Z1,Z2,Z3,Z4,Z5,Z6,Z0[1:2];
   'PROCEDURE'MM(N,Z1,Z2,Z3);
   'VALUE'N;'INTEGER'N;'ARRAY'Z1,Z2,Z3;
   'BEGIN'
     ..... }过程体
   'END';
   INPUTR(Z1,Z2,Z3,Z4,Z5);
   MM(1,Z1,Z2,Z0); PUTAR(Z0,2);
   MM(2,Z1,Z2,Z0); PUTAR(Z0,2);
   MM(3,Z1,Z2,Z0); PUTAR(Z0,2);
   MM(4,Z3,Z4,Z0); PUTAR(Z0,2);
   MM(5,Z5,Z0,Z6); PUTAR(Z0,2); PUTAR(Z6,2);
'END'
******
```

1.2 复n次方程求全部根的 Muller 方法

一、功能

对于系数为复数（或实数）的 n 次代数方程

$$a_n x^n + a_{n-1} x^{n-1} + \cdots + a_1 x + a_0 = 0$$

利用*Muller*方法可以求出它的全部根。如果方程系数为实数，使用本程序时只需要将存放系数虚部的位置填 0 即可。如果所给方程的系数有$a_0 = a_1 = a_2 = \cdots = a_K = 0$ $(K < n)$，则首先进行降阶处理，使新方程不含有零根时再调用本过程。

二、方法简述

令 $F(x) = a_n x^n + a_{n-1} x^{n-1} + \cdots + a_1 x + a_0$

利用递推公式

$$\lambda = \frac{-2F(x_i)\delta_i}{g_i \pm \sqrt{g_i^2 - 4F(x_i)\delta_i\lambda_i(F(x_{i-2})\cdot\lambda_i - F(x_{i-1})\delta_i + F(x_i))}}$$

其中 $g_i = F(x_{i-2})\lambda_i^2 - F(x_{i-1})\cdot\delta_i^2 + F(x_i)(\lambda_i + \delta_i)$

$$\delta_i = 1 + \lambda_i \qquad \lambda = \lambda_{i+1} = \frac{x_{i+1} - x_i}{x_i - x_{i-1}}$$

取初值$x_0 = -1$，$x_1 = 0$，$x_2 = 1$。

由上述迭代关系式，求出序列x_3，x_4，…，x_i，….

当x_i满足$\dfrac{|x_i - x_{i-1}|}{|x_{i-1}|} < \varepsilon$时，认为$x_i$就是原方程 $F(x)=0$ 的一个根（暂记作x^*）。

方程的降阶。在求出x^*后，利用公式

$$\begin{cases} b_i = a_{i+1} + b_{i+1} x^* & \text{当 } i = n-2,\ n-3,\cdots,1,0\text{时} \\ b_{i+1} = 0 & \text{当 } i = n-1 \text{ 时} \end{cases}$$

使原方程分离出已经求出的根x^*，降阶一次，反复运用以上步骤，直到求出全部根。

三、过程参数说明

过程名称　$FXY(N, A, EPS, C)$；

N：所求方程的次数，整型值参。

$A[0:N, 1:2]$：方程$F(x)=0$的$n+1$个系数，使用时应当注意顺序与次数的统一。

$A[i, 1]$存实部，$A[i, 2]$存虚部，实型数组。

EPS：根的相对误差限，实型值参。

$C[1:N, 1:2]$：存放$F(x)=0$的全部根，$C[i,1]$存实部，$C[i, 2]$存虚部，实型数组。

四、过程

```
'PROCEDURE'FXY(N,A,EPS,C);
'VALUE'N,EPS;'INTEGER'N;'REAL'EPS;'ARRAY'A,C;
'BEGIN'
  'INTEGER'J,J1,J2,J3;
  'ARRAY'X0,X1,X2,X3,X4,X5,X6,X7,X8,X9[1:2],X10,X11,
  X12,X13,X14,X15,X16,X17,X18,Z1,Z2,G[1:2],Z3,Z4,Z5,
  Z6,Z7,Z8,Z9,A1,A2,F0 F1,F2[1:2],B[0:N,1:2],
  F[0:2,1:2];
  'SWITCH'SW:=L,L1,L2,L3,L4,L5,L6;
  F[0,1]:=-1;F[2,1]:=1;F[0,2]:=F[1,1]:=F[1,2]:=F[2,2]:=0;
  X0[1]:=X1[1]:=1;  X0[2]:=X1[2]:=0;  J2:=N;
L1:'FOR'J1:=0,1,2'DO'
  'BEGIN'
    A1[1]:=A[N,1];    A1[2]:=A[N,2];
    X5[1]:=F[J1,1];   X5[2]:=F[J1,2];
    'FOR'J3:=N-1'STEP'-1'UNTIL'0'DO'
    'BEGIN'
```

```
    A2[1]:=A[J3,1];      A2[2]:=A[J3,2];
    MM(3,A1,X5,X3);      MM(1,X3,A2,A1);
  'END';
  B[J1,1]:=A1[1];        B[J1,2]:=A1[2]
'END';
F0[1]:=B[0,1];  F0[2]:=B[0,2];  F1[1]:=B[1,1];
F1[2]:=B[1,2];  F2[1]:=B[2,1];  F2[2]:=B[2,2];
MM(3,X1,X1,X2);  MM(1,X1,X0,Z9);
MM(1,Z9,X1,Z8);  MM(3,F0,X2,Z7);
MM(3,F2,Z8,Z6);  MM(3,Z9,Z9,Z5);
MM(1,Z6,Z7,Z8);  MM(3,F1,Z5,Z7);
MM(2,Z8,Z7,G);   MM(3,F0,X1,Z8);
MM(1,Z8,F2,Z7);  MM(3,F1,Z9,Z8);
MM(2,Z7,Z8,Z6);  X7[1]:=-4;
X7[2]:=0;        MM(3,X7,Z6,Z8);
MM(3,Z8,X1,Z6);  MM(3,Z6,F2,Z7);
MM(3,Z7,Z9,Z8);  MM(3,G,G,Z7);
MM(1,Z7,Z8,Z6);  MM(5,Z6,Z4,Z5);
MM(1,G,Z4,Z6);   MM(1,G,Z5,Z7);
X8[1]:=SQRT(Z6[1]↑2+Z6[2]↑2);
X8[2]:=SQRT(Z7[1]↑2+Z7[2]↑2);
'IF'X8[1]'GR'X8[2]'THEN''GOTO'L1'ELSE''GOTO'L2;
L1: MM(3,Z6,X0,X2);  'GOTO'L3;
L2: MM(3,Z7,X0,X2);
L3: X11[1]:=-2;  X11[2]:=0;
  MM(3,F2,X11,Z8);  MM(3,Z8,Z9,X11);
  'IF'SQRT(X2[1]↑2+X2[2]↑2)'LS'10-10'THEN'
  X12[1]:=10'ELSE'X12[1]:=1;  X12[2]:=0;
  MM(3,X12,X11,X13);  MM(3,X12,X2,X14);
  MM(4,X13,X14,X1);
  F[0,1]:=F0[1]:=F[1,1];  F[0,2]:=F0[2]:=F[1,2];
```

```
    F[1,1]:=F1[1]:=F[2,1];  F[1,2]:=F1[2]:=F[2,2];
    MM(2,F1,F0,Z8);  MM(3,X1,Z8,Z9);
    MM(1,Z9,F1,F2);  MM(2,F2,F1,Z4);
    F[2,1]:=F2[1];  F[2,2]:=F2[2];
    X3[1]:=SQRT(Z4[1]↑2+Z4[2]↑2);
    X3[2]:=SQRT(F2[1]↑2+F2[2]↑2);
    'IF'X3[1]/X3[2]'GR'EPS'THEN''GOTO'L;
    'FOR'J:=J2'STEP'-1'UNTIL'0'DO'
    'BEGIN'
      'IF'J=J2'THEN''GOTO'L4'ELSE''GOTO'L5;
L4:  B[J,1]:=B[J,2]:=0;'GOTO'L6;
L5:  B[J,1]:=A[J+1,1]+B[J+1,1]*F2[1]-B[J+1,2]*F2[2];
     B[J,2]:=A[J+1,2]+B[J+1,1]*F2[2]+B[J+1,2]*F2[1];
L6:
    'END';
    'FOR'J:=N'STEP'-1'UNTIL'0'DO'
    'BEGIN'
      A[J,1]:=B[J,1];A[J,2]:=B[J,2]
    'END';
    C[J2,1]:=F2[1];  C[J2,2]:=F2[2];
    J2:=J2-1;  X1[1]:=1;  X1[2]:=0;
    F[0,1]:=-1;  F[2,1]:=1;
    F[0,2]:=F[1,1]:=F[1,2]:=F[2,2]:=0;
    'IF'J2'GR'1'THEN''GOTO'L;
    A1[1]:=-A[0,1];  A1[2]:=-A[0,2];
    A2[1]:=A[1,1];  A2[2]:=A[1,2];  MM(4,A1,A2,F2);
    C[1,1]:=F2[1];  C[1,2]:=F2[2]
  'END';
```

注：本过程使用的复数基本运算过程*MM*请参考本书关于复数基本运算过程的介绍。

五、例题

1.求六次勒让德多项式 $231x^6-315x^4+105x^2-5=0$ 的根.

计算结果：$x_{1,2}=\pm 0.932469514$；

$x_{3,4}=\pm 0.661209387$；$x_{5,6}=\pm 0.238619186$；

2.求 $x^3-3x^2-10x+24=0$ 的根.

计算结果 $x_1=-3$；$x_2=4$；$x_3=2$.

3.求 $x^3-2x-5=0$ 的根.

计算结果 $x_1=2.09455148$；

$x_{2,3}=-1.04727547\pm 1.13593989i$.

4.求 $x^6-(4.1+6i)x^5-(4.8-12.3i)x^4+4x^2-(16.4+24i)x-19.2+49.2i=0$ 的全部根.

计算结果 $x_1=0.21_{10}1+0.299999999_{10}1i$；

$x_2=-1.0-1.0i$； $x_3=2.0+3.00000001i$；

$x_4=1.0-1.0i$； $x_5=-1.0+1.0i$；$x_6=1.0+1.0i$.

5.求 $x^{41}+x^3+1=0$ 的全部根.

计算结果

$x_{1,2}=1.01430467\pm 0.809230298_{10}-1i$

$x_{3,4}=0.987183858\pm 0.240354384i$

$x_{5,6}=0.933664045\pm 0.392546384i$

$x_{7,8}=0.855158028\pm 0.532633535i$

$x_{9,10}=0.753719530\pm 0.655380276i$

$x_{11,12}=0.632339827\pm 0.753401200i$

$x_{13,14}=0.507568639\pm 0.810573439i$

$x_{15,16}=0.417151578\pm 0.871067309i$

$x_{17,18}=0.289812131\pm 0.946423675i$

$x_{19,20}=0.139165435\pm 0.992476734i$

$x_{21,22}=-0.0197286321\pm 01.00935216i$

$x_{23,24}=-0.180206043\pm 0.997962233i$

$x_{25,26}=-0.336984323\pm 0.959227613i$

$x_{27,28}=-0.485280239\pm 0.894538370i$

$x_{29,30}=-0.620672505\pm 0.805889307i$

$x_{31,32}=-0.739100565\pm 0.695904353i$

$x_{33,34}=-0.836863005\pm 0.567825669i$

$x_{35,36}=-0.910511135\pm 0.425528151i$

$x_{37,38}=-0.956339011\pm 0.273776209i$

$x_{39,40}=-0.968140341\pm 0.120866955i$

$x_{41}=-0.952483875.$

程序

```
'BEGIN'
  'INTEGER'N;  READI(N);
  'BEGIN'
    'REAL'EPS;  'ARRAY'A[0:N,1:2],  C[1:N,1:2];
    'PROCEDURE'MM(N,Z1,Z2,Z3);
    'VALUE'N;  'INTEGER'N;  'ARRAY'Z1,Z2,Z3;
    'BEGIN'
      ..... }过程体
    'END';
    'PROCEDURE'FXY(N,A,EPS,C);
    'VALUE'N,EPS;  'INTEGER'N;  'REAL'EPS;
    'ARRAY'A,C;
    'BEGIN'
      ..... }过程体
    'END';
    INPUTR(A);  EPS:=10-10;
```